
Abstract

We introduce Obol1 an imperative Object-oriented variant of ISWIM2. The Obol inter-
preter is written in Java, and Obol and Java objects can easily be used together in the
same (Obol, or Java) application.

1An Obol is (the weight of) half a scruple of silver: Charon’s price for the one-way ferry-ride across the river
Styx. Obol is not to be confused with Cobol “A weak, verbose, and flabby language used by card wallopers
to do boring mindless things on dinosaur mainframes.” (Jargon: April 2001) “The use of COBOL cripples the
mind; its teaching should, therefore, be regarded as a criminal offense.” (E.W.Dijkstra). The acronym stands for
“COmmon Business-Oriented Language.”

2ISWIM was a programming language devised by Peter J. Landin and described in his highly influential
article, The next 700 programming languages, CACM 9(3):157-166 (Mar 1966). The acronym stands for “If you See
What I Mean”.

1 Tutorial Introduction

1.1 The Interactive Obol System

The interactive Obol system is started with the Unix/Linux command3

obol

The system repeatedly

• Prompts its user with the text >>, then

• reads an Obol phrase followed by a semicolon, then

• evaluates the phrase in the current context, and finally

• prints the resulting value, if any.

The remainder of this tutorial is presented as the transcript of a session with the Obol system
in which phrases input by the user are prefixed with >> and system responses are printed
below them, without prefix.

Here, for example, we show an interaction in which the Obol interpreter evaluates 303:

>> 30**3;

27000

1.2 Declarations

It is a simple matter to define local procedures and variables. For example, the following
expression causes the phrase n3 to be evaluated in a context in which the name n is associated
with a location containing 30.

>> let n = 30 in n**3;

27000

In the following example, we associate the name x with (a location containing) 10, and the
name f with an Obol procedure that computes the value of the factorial function at n, then
evaluate the phrase f x, yielding 3628800.

3If you only have the Obol.jar file, use java -jar Obol.jar, or (Windows) double-click on the Obol.jar file.
For more information about the interactive Obol window see Appendix F. For more information about installing
Obol see Appendix G.

>> let f(n) = if n==0 then 1 else n*f(n-1) fi

and x = 10

in f(x);

3628800

The scope of the collection of assocations made by such a local declaration is the text of the Scope
Rulesbody of the phrase, as well as the bodies of all functions defined by the same declaration.

Furthermore, in a declaration of the form

let p1 = e1 and p2 = e2 · · · and pn = en

the scope of the association made by pi = ei includes ei+1, ei+2, · · · en.

>> let x=1

in let y=x and z=y+1 and x=x+1

in [x, y, z];

[2, 1, 2]

Scopes nest properly, as the following annotated example shows. Overbraces name associ- Nested
Scopes

ations, underbraces show the scope of named assocations.

>> let

a1︷    ︸︸    ︷
x = 10

and

a2︷                                                            ︸︸                                                            ︷
y = let

a3︷     ︸︸     ︷
x = x − 7︸︷︷︸

a1

and

a4︷     ︸︸     ︷
y = x + 1︸︷︷︸

a3

in x + y︸︷︷︸
a3,4

in x ∗ y︸︷︷︸
a1,2

;

70

To declare functions and variables persistently in the top-level context we use a special form
Persistent

Declarations

of let phrase – one without an explicit body. In this case the scope of the collection of
associations is all subsequent expressions.

In the following example we introduce initialised variables x and y by associating the name
x with a location initially containing the value 3, and the name y with a location initially
containing the value 4. Declarations that appear at the top-level like this are executed for
their effect on the context in which subsequent expressions are evaluated. They have no
value, so the system prompts for the next phrase immediately.

>> let x = 3 and y = 4;

>> x;

3

>> y;

4

>> x==4;

true

>> x==y;

false

The value of a variable can be changed with an assignment. Assignments are executed for
Assignm

ent

their effect on variables and, like all other Obol expressions executed for their effects, yield
the value {} (pronounced “null”, or “nil”).4

>> x:=42;

{}

>> x;

42

The variable to be assigned to can be determined by an expression, for example:

>> if x<y then x else y fi := 99;

>> (x, y);

(42, 99)

Simultaneous assignments are executed by first evaluating all the right-hand sides, then Sim
ultaneous

Assignm
ent

simultaneously updating the locations associated with the variables.

>> x, y := y, x;

>> (x, y);

(99, 42)

There is a special case – the unpacking assignment: if there is only one expression on the right
hand side, and more than one on the left, then the right hand expression must result in a
tuple of the right length.

>> let rhs = (y, x);

>> x, y := rhs;

>> (x, y);

(42, 99)

4In a normal interactive session with Obol we can suppress the printing of {} by making the assignment:
system.PrintNil := false; Henceforth we shall assume this has been done.

There is a corresponding special case for a declaration – the unpacking declaration: if there
is only one expression on the right hand side, and more than one on the left, then the right
hand expression must result in a tuple of the right length.5 .

>> let rhs = (y, x);

>> let a, b = rhs;

>> (a, b);

(42, 99)

This can be convenient when declaring several variables at once:

>> let a, b = (101, 102);

>> (a, b);

(101, 102)

BUT NOT:

>> let a, b = 101, 102;

Syntax Error: misplaced comma

5See Note 1 on page i for a more general account of unpacking declarations.

A sequential expression is written in the form {e1; e2; · · · en} and is evaluated by executing its
Sequencing

component expressions one by one. It yields the value of the final expression.

>> let x=3 in { x:=x+2; x };

5

An iteration always yields null, whatever the value of the expression that is being iterated.

>> let x=3 in while x>0 do { x:=x-1; 999 };

{}

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

The classic “Hello World” program is:
Hello, W

orld!>> println("Hello, World");

Hello, World

Here the effect of evaluating the print expression is to output the text Hello, World. It then
yields the value {} – which is printed by the evaluator only if system.PrintNil is true.6

6In note 2 (page i) we show how to make an independently runnable program.

Many of the remaining Obol control structures will be familiar. In the following example, Iteration
andChoice

which demonstrates the use of iteration and choice expressions, we define and use a proce-
dure that computes the greatest common divisor of its arguments. Notice that the body of
the procedure is a sequential expression whose last component is the expression denoting
the result – Obol has nothing like a return command.

>> let gcd(m, n) =

{ while m!=n do

if

m>n then m:=m-n

else if

m<n then n:=n-m

fi;

m

};

>> gcd(36, 15);

3

>> gcd(121, 13);

1

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

An alternative way of defining the same procedure demonstrates the conditional expression
and continued relation notations.7

>> let gcd(m, n) = 0<m<n => gcd(m, n%m),

m>n>0 => gcd(m%n, n),

m==0 => n,

n==0 => m,

m<0 => -gcd(-m, n),

n<0 => -gcd(m, -n),

n;

The conditional expression notation g⇒ e1, e2 means the same as if g then e1 else e2 fi.
For any relations ≺

1
,≺

2
, · · · ≺

n
the “continued relation” notation e1 ≺

1
e2 ≺

2
· · · ≺

n−1
en means

the same as e1 ≺
1
e2 && e2 ≺

2
e3 && · · · en−1 ≺

n−1
en.

7Yes, it’s not the most efficient way!

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

Type annotations can be associated with any variable or expression, as a way of conveying
Type

Annotations

the programmer’s intentions about an invariant property (albeit weak) that is expected to
hold of that variable or expression.

A type expression is written in the same superficial syntax as an ordinary expression8, but at
present type expressions are completely ignored by Obol – indeed they don’t make it past the parsing
stage.

We expect that type annotations will eventually form the basis for a proper (static) typechecker
for Obol.

Examples:

>> let factorial:int->int = \(n:int) -> n==0=>1, n*factorial(n-1);

>> let f(x:int) = case factorial x of (n:int)->n;

>> let gcd(m:int, n:int):int =

0<m<n => gcd(m, n%m),

m>n>0 => gcd(m%n, n),

m==0 => n,

n==0 => m,

m<0 => -gcd(-m, n),

n<0 => -gcd(m, -n),

n;

8Except that the symbol ->may be used as an infix

Case expressions make it possible to discriminate on the structure and content of values.
Case Expressions

>> let greet name =

case n of "bernard" -> "What-ho!"

or "jeff" -> "G’day!"

or (first, second) -> "Hello "++first++" "++second

or _ -> "Greetings, esteemed "++name;

A case expression takes the form

case expression0 of

pattern1->expression1
or ...

or patternn->expressionn

First expression0 is evaluated, yielding value v. Then the patterns are matched9 against v
in order of their occurence. If none matches then the case expression fails – throwing a
bind exception. Otherwise the expression corresponding to the first pattern that matches
is evaluated in the current context, augmented by the result of binding the variables in the
pattern to the values at the corresponding positions in v.

The patterns can be arbitrarily complex, and can contain repeated occurences of the same
variable. The “variable” _means “don’t care” in a pattern, and never gets bound.

Sequence display patterns match any sequence of the right size, as do tuple displays:

>> case [1,2,3] of []->0 or [_]->1 or [_,_]->2 or other->math.maxint

9223372036854775807

>> case [1,2] of []->0 or [_]->1 or [_,_]->2 or other->math.maxint

2

>> case (1,2,3) of ()->0 or (_)->1 or (_,_)->2 or other->math.maxint

1

>> case (1,2) of ()->0 or (_,_)->1 or (_)->2 or other->math.maxint

1

A match where there are repeated occurences of the same variable in a pattern will only
succeed if the occurences of the variable are all bound to equivalent values.

>> case [1,2,3] of [x,x,x] -> true or other -> false;

false

>> case [[1],[1],[1]] of [x,x,x] -> true or other -> false;

true

Record patterns match any record containing at least the fields in the pattern.

>> case {| a=3; b=4 |} of {| a=x |} -> x;

3

9See Appendix E for pattern-matching rules.

It can be useful to “hide” variables, so that they cannot inadvertently be referenced or
Hiding

variables
changed. Here is a very simple example, in which the scope of the declaration of hidden is
the declaration below. That declaration defines two procedures that communicate via the
hidden variable: clients of the procedures cannot change the variable because they cannot
“see” it.

>> let hidden = 0

in let nextNumber() = { hidden:=hidden+1; hidden }

and reset() = { hidden:=0 };

>> nextNumber();

1

>> nextNumber();

2

>> hidden;

Error: variable undeclared: hidden

>> reset();

>> nextNumber();

1

The formal parameter(s) of a function are actually patterns. An attempt to apply a function Param
eters

arePatterns

to an argument that doesn’t match its parameter throws an exception.

>> let f(a,b,c) = a+b+c;

>> f[1,2,3];

Error: (a, b, c) does not match [1, 2, 3] at f[1, 2, 3]

>> let f[a,b,c] = a+b+c;

>> f[1,2,3];

6

Parameters are passed by value:

>> let f x y = x:=y;

and x=3

in { f x 55; x };

3

Functions are first-class values, so we can define higher-order functions. For example:

>> let compose f g x = f(g x)

let pair x y = (x, y)

let both f (x, y) = (f x, f y)

let par (f, g) x = (f x, g x)

let add1 x = x+1;

>> par (add1, compose add1 add1) 3;

(4, 5)

>> compose (both (par (add1, compose add1 add1))) (pair 5) 4;

((6, 7), (5, 6))

The “lambda expression” written \pat → expr means “that function of pat whose value is
Lam

bda

Expressions

expr.”

>> both (\ x -> x+1) (3, 4)

(4, 5)

>> (\f -> \g -> \x -> f(g x)) add1 add1 3

5

1.3 Exceptions

The evaluation of a throw expression results in the throwing of a “catchable” exception – as
Exceptions

does the occurence of various other forms of exceptional condition such as division by zero.
If such an exception is not caught, then the Obol system reports the dynamic context10 in
which it occured.

>> let f(n) =

if

n<0 then n

else if

n==0 then throw "Non-negative"

else

f(n-1)

fi;

>> f(3);

Error: Non-negative

at throw("Non-negative")

at f(n-1)

at f(n-1)

at f(n-1)

at f(3)

The exceptions raised during the execution of an expression can be caught in a (dynamically
enclosing) try block. When no exception is raised the value of the try block is the value of
the expression being tried:

>> try f(-3) catch s -> { println ("Caught: "++s); 42 };

-3

Normally an exception is caught and acted on in the catch clause of the closest dynamically-
enclosing try block.11 In the following example the exception is not reported, and the result
of the try block is the value of the catch expression.

>> try f(3) catch s -> { println ("Caught: "++s); 42 };

Caught: Non-negative

42

A then clause may follow the catch clause. In this case the value of the try block is obtained
by evaluating the then expression – whether or not an exception was thrown during the
evaluation of the expression being tried.

10i.e. the call stack
11See Note 3 on page ii for a more detailed account.

>> try println(f(3))

catch s -> { println ("Caught: "++s); 42 };

then println "Finished";

Caught: Non-negative

Finished

>> try println(f(-3))

catch s -> { println ("Caught: "++s); 42 };

then println "Finished";

-3

Finished

The catch clause may be omitted if there is a then clause. In this case the value of the try
block is obtained by evaluating the then expression after the expression being tried has been
evaluated to completion or raised an exception.

>> try { print 1; print 2; throw "fail"; print 4 }

then println 3;

123

1.4 Sequences and Mappings

1.4.1 Sequences

The only predefined data structures are finite sequences and finite mappings. Finite sequences
Sequences

behave like functions from an initial subsequence of the natural numbers.12 A sequence display
constructs an immutable sequence – one whose elements may not be changed.

>> ["the", "rain", "in", "spain"];

["the", "rain", "in", "spain"]

>> ["the", "rain", "in", "spain"](0);

"the"

>> let con = ["the", "rain", "in", "spain"];

>> #con;

4

>> con(1) := "brain";

Error: ["the", "rain", "in", "spain"] is not a mutable seq.

at con(1):="brain"

12The precise representation of a sequence depends on how it was constructed, as does the efficiency of the
various operations on it. See appendix B for details.

A for loop iterates over a sequence:

>> for word in con do { print word; print " " };

the rain in spain

The elements of mutable sequences (called arrays) can be changed. The built-in procedure
newArray constructs an array of a specific length from a function or int-function-like value.13

In the following examples we first construct an array from a function that doubles its argu-
ment, and then construct one from a sequence.

>> let arr = newArray(3, \i -> i*2);

>> arr;

[0, 2, 4]

>> arr(2) := 99; arr;

[0, 2, 99]

>> let var = newArray(#con, con);

>> var(1) := "brain"; var;

["the", "brain", "in", "spain"];

An immutable sequence can be constructed using the built-in procedure newTable.

>> let ro = newTable(3, \i -> i*2);

>> ro;

[0, 2, 4]

>> ro(2) := 99;

Error: [0, 2, 4] is not a mutable seq.

at ro(2)

at ro(2):=99

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

A range is a read-only subsequence of the integers, which is represented in a concise form:14
Range

>> let twenties = 20 # 10;

>> #twenties;

10

>> for i in twenties do print(i, " ");

20 21 22 23 24 25 26 27 28 29

>> for i in 0 # #twenties while i<4 do print(twenties(i), " ");

20 21 22 23

13i.e. any value or object for which the application operator () has type consistent with int->val.
14The infix # operator is pronounced “for”.

>> for i in 0 # #con do println (i, " ", con(i));

0 the

1 rain

2 in

3 spain

A range with negative length runs in the opposite direction

>> newTable(6, 10#-6);

[10, 9, 8, 7, 6, 5]

>> for i in 10#-6 do print(i, " ");

10 9 8 7 6 5

>> #(10#-6)

6

Ranges with matching endpoints behave appropriately when catenated, otherwise they be-
have just like the sequences they represent:

>> (0#5) ++ (5#5);

0#10

>> (10#-5) ++ (5#-5);

10#-10

>> (10#5) ++ (15#5);

10#10

>> (10#5) ++ (15#-5);

[10, 11, 12, 13, 14, 15, 14, 13, 12, 11]

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

Hitherto we have worked only with simple types, for which the notion of equality is utterly Sequence

Identity
and

Equivalence

straightforward. When comparing the value of (1 + 4) with the value of (3 + 2) it would be
foolish to try to distinguish between these 5s because they came about in different ways – a
5 is a 5 is a 5! There is, therefore, just a single equality on the simple types (such as number),
and two elements of a simple type are equivalent if they are equal.

Once we start to work with data structures that can represent more complicated values,
we have to take a more refined view, and consider two notions of equality. One is that of
equivalence-as-a-value, the other is that of identical-address-in-the-computer. In Obol we
use the symbol == to mean “identical” and the symbol === to mean “equivalent”.15

15As a rule of thumb, the user of a (composite) data type will mostly be concerned with equivalence, whilst its
implementer may also need to be concerned with identity.

Two sequences are equivalent if they contain equivalent elements in the same order.

>> [1, 2, 3] === newTable(3, \i->i+1);

true

>> [1, 2, 3] === [1, 2] ++ [3];

true

>> [1, 2, 3] === 1 # 3;

true

>> [[1], [1,2], [1,2,3]] === [1#1, 1#2, 1#3];

true

Equivalent sequences are not necessarily identical. On a particular processor the identity of a
sequence is a function of its elements in order and the exact moment it was constructed.

>> [1, 2, 3] == 1 # 3;

false

>> let s = [1, 2, 3];

>> s == s;

true

>> s == [1, 2] ++ [3];

false

Sequences need not necessarily be composed of elements of the same type.16
Typesand

Sequences>> ["fun", ’f’, 23];

["fun", ’f’, 23]

They can even contain references to themselves! The Obol system prints cyclic data structures
specially: printing an ellipsis if it reaches a point in the structure that it has started, but not
yet finished, printing.

>> let loop = newArray(2, 33) in { loop(1):=loop; loop };

[33, ...]

Tools like those familiar from functional programming, such as zip, map, filter can easily
be built:17

16To be precise, the Obol interpreter does not require sequences to be homogeneously typed, but some opti-
mising compilers for Obol do have such a requirement.

17Indeed they can be built in many slightly different variants, depending on the precise mutabilities required.

>> let MAP f seq = newTable(#seq, \i -> f(seq i))

and FILTER p seq = cat(newTable(#seq, \i -> p(seq i)=>[seq i],[]))

and ZIP seqs = newTable(#(seqs 1), \i -> newTable(#seqs, \j -> seqs j i));

>> FILTER odd (0#10) where odd n = n%2==1;

[1,3,5,7,9]

>> MAP bit (0#10) where bit n = n%2;

[0, 1, 0, 1, 0, 1, 0, 1, 0, 1]

>> ZIP [[1,2],[3,4], [5,6,7]];

[(1, 3, 5), (2, 4, 6)]

There’s also a built-in filter: it selects the subsequence of elements of its second argument
which satisfy the predicate which is its first argument.

>> filter(\i->i%3==0, 0#50);

[0, 3, 6, 9, 12, 15, 18, 21, 24, 27, 30, 33, 36, 39, 42, 45, 48]

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

1.4.2 Mappings

A finite mapping display constructs an immutable finite mapping from values to values.
M

apping
>> let mapping = [| "foo" -> "a mysterious prefix", "baz"->"a curious suffix" |];

>> mapping("foo");

"a mysterious prefix"

>> for word in mapping do println(word, " is ", mapping word);

foo is a mysterious prefix

baz is a curious suffix

>> (mapping.has("fig"), mapping.has("foo"));

(false, true)

Applying a mapping to a key outside its domain causes a ‘subscript exception to be thrown.

>> [|1->2|](3);

Error: key outside domain

at ([| 1->2 |])(3)

>> try [|1->2|](3) catch e->e;

‘subscript ([|1->2|], 3)

The catenation operator takes the union of its arguments, except that when there is conflict
between its left and right operand it prefers the right.

>> mapping ++ [| "foobaz" -> "an odd couplet", "foo" -> "my default identifier" |];

[|"foo"->"my default identifier", "baz"->"a curious suffix",

"foobaz"->"an odd couplet"|]

The newDict function constructs a new mutable mapping.

>> let d = newDict();

>> d "foo" := "baz";

>> d "baz" := "foo";

>> d;

[|"foo"->"baz", "baz"->"foo"|]

>> d.update([| "foo"->24, "pig"-> 23, "nap"->"won"|]); d;

[|"foo"->24, "pig"->23, "nap"->"won", "baz"->"foo"|]

If m is a mapping (mutable or not) then m.keys() is a sequence containing all the keys (domain
elements) of the mapping in an arbitrary order, and m.values() is a sequence containing all
the values (range elements) of the mapping in an arbitrary order.18

It is unwise (though not forbidden) to use mutable structures as keys in mappings.

18The more mathematically oriented may prefer the synonyms: m.dom() = m.keys() and m.ran() = m.values().

1.5 Records

A record expression takes the form:
Records

{| id1 = expr1; · · · idn = exprn |}

The expressions are evaluated in turn, and their values associated with the field names
id1, · · · idn.

The field idi of a record r is selected using the conventional notation r.idi

For example:

>> let ratio = {| num=5; den=6 |};

>> ratio.num;

5

>> ratio.den;

6

Individual fields can be assigned to:

>> ratio.den:=50;

>> ratio;

{| den = 50; num = 5 |}

On a given machine the identity of a record is a function of the exact moment it was created Record
Identity

and the values associated with its fields.

>> {| num=5; den=6 |} == {| num=5; den=6 |};

false;

>> let rat = {| num=5; den=6 |};

>> rat==rat;

true

>> rat == {| num=5; den=6 |};

false

By default, two records are equivalent if they have the same field names and the values of
Record

Equivalence

corresponding fields are themselves equivalent.

>> rat === {| num=5; den=6 |};

true

>> rat === {| num=5; den=6; other={} |};

false

Records are frequently used to represent types of value that are somewhat richer in structure
than the natural equivalence recognises, so it is useful for us to be able to define specific
equivalences for specific types. Obol supports this, and the matter of structural equivalence
is taken up at length in Notes 5, and 6 (page iii).

Records are extended using the ++ operator.
Record

Extension
The result of extending a record r with fieldnames f1, · · · , fn and a record s with fieldnames
g1, · · · , gm is a record with fieldnames f1, · · · fn ∪ g1, · · · gm. Field f of the result has value s.f if f
is one of g1, · · · , gn, otherwise it has value r.f .19

For example:

>> let rat = {| num=5; den=6 |};

>> let tat = rat ++ {| num=7; kind="slow" |}

>> tat;

{| den=6; kind="slow"; num=7 |}

Storage for the fields of the result of an extension is shared with storage for the fields of the
operands from which they came.

>> tat.num, tat.den := 66, 77;

>> tat;

{| den=77; kind="slow"; num=66 |}

>> rat;

{| den=77; num=5 |}

19When a field f appears in both operands of a extension, we say that the right hand operand “overrides f ”.

1.6 Data Constructors

A data constructor (or atom) is a (mnemonically-named) constant, written as a backquote
followed by a name.

Data Constructors are intended to be used as mnemonic tags in data structures. Those familiar
with programming in Haskell, Caml, or Standard ML will recognise data constructors as
being akin to the constructors found in those languages.

A data constructor just stands for itself.

>> ‘cons;

‘cons

A data constructor may also be “applied” to a value – in which case the result is a construction
– in which the value (known as the content) is tagged by the data constructor.

>> ‘cons(1,2);

‘cons(1,2)

Two constructions are equivalent if they are identical, or if they have the same data constructor
as tag, and equivalent content.

>> ‘tag[1,2] === ‘tag(1#2);

true

Data Constructors may appear anywhere in patterns, and a data constructor matches only a
data constructor formed from exactly the same name. A pattern of the form ‘name1(pattern)
matches a value of the form ‘name2(val) if, and only if, name1 is identical to name2 and pattern
matches val.

For example, below we show how to use data constructors in defining a collection of functions
to manipulate lists.

>> let cons(h, t) = ‘cons(h, t)

and hd(‘cons(h,_)) = h

and tl(‘cons(_,t)) = t

and nil = ‘nil

and null s = case s of ‘nil->true or ‘cons _ -> false

and len s = case s of ‘nil->0 or ‘cons (_, t) -> 1+len t;

1.7 Modules

A module is defined by a module expression. For example, here we define a module that Defining
M

odules
implements a simple random number generator together with various ways of using it.

let random =

module

let next(bitCount) =

{ seed := (seed * 0x5DEECE66D + 0xB) & ((1 << 48) - 1);

seed >> (48-bitCount);

}

and seed = 253

and nextByte () = next(8)

and nextInt () = next(32)<<32+next(32)

and nextBool () = next(1)==1

end;

The variables defined by a module can be used without importing it.

>> newTable(5, \i->random.nextByte());

[239, 37, 127, 20, 29]

A module can be imported into a limited scope Im
porting

M
odules

>> let randBytes n = import random in newTable(n, \i->nextByte());

>> randBytes(10);

[44, 101, 221, 93, 253, 36, 145, 74, 239, 134]

... or into the global context:

>> import random;

>> newTable(5, \i->random.nextBool());

[false, true, false, true, false]

The global procedure loadModule can be used to load Obol code from a file. It wraps the Loading
M

odule
Code

top-level declarations in the file up as a module, and returns that module.20

For example, suppose that the code to implement the random number generator is present in
the file rand.ob then loading the file yields a module that has a single binding in it, namely
that of random:

>> loadModule "rand.ob";

module

random=module

next=fun next(bitCount)->val

20If a file has already been loaded this way, then its module value is used again, rather than being reconstructed.
To force the reloading of a loaded module it must be unloaded using the library procedure system.unload

nextBool=fun nextBool()->val

nextByte=fun nextByte()->val

nextInt=fun nextInt()->val

seed=253

end

end

In order for the module to be any use to the programmer it must be imported.

>> import loadModule("rand.ob");

>> newTable(5, \i->random.nextBool());

[false, true, false, false, false]

Obol’s import construct provides a shortcut for importing a module directly from a file – just
use a string-valued expression in place of a module-valued expression.

>> import "rand.ob";

>> newTable(5, \i->random.nextBool());

[false, true, false, false, false]

Modules are first-class values, in the sense that they can be computed as the value of any
expression. This makes it easy to define parameterised modules.

It would, arguably, be better to parameterise our random number-generating module with
an initial value for the seed, and to “hide” the workhorse procedure next. The following
definition accomplishes this, incidentally adding a generator for reals in the range 0.0-1.0.

let random(seed)=

let next(bitCount) =

{ seed := (seed * 0x5DEECE66D + 0xB) & ((1 << 48) - 1);

seed >> (48-bitCount);

}

and real2E53 = global.math.real(1<<53)

in

module

let nextByte () = next(8)

and nextInt () = next(32)<<32 + next(32)

and nextBool () = next(1)==1

and nextReal () = (next(26)<<27+next(27)) / real2E53

end;

If this code is placed in the file random.ob then we can have several random number generator
modules in action at once.

>> let r333 = import "random.ob" in random(333)

and r444 = import "random.ob" in random(444);

>> for t in 0#5 do println(r333.nextReal(), "\t", r444.nextReal());

0.029830581147327173 0.03977411315859547

0.5833470435760478 0.7639190478719554

0.4555609068093781 0.1525471094440337

0.59304481815736 0.7599605456158061

0.2466127280767071 0.15323355626186452

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

As a second example, the following defines a sorting utility module parameterised by a
module that defines a precedes relation.

let sorting(order) =

// Requires order: module precedes:(val,val)->bool end

module

let bubble seq =

// Returns an array that permutes seq with elements ordered

// consistently with order.precedes

{ worker (isArray seq=>seq, newArrayOf(seq))

where worker a =

{ for i in 0#(#a) do

for j in (i+1)#(#a-i-1) do

if order.precedes(a j, a i) then a j, a i := a i, a j fi;

a

}

}

and sort seq =

// Returns a table that permutes seq with elements ordered

// consistently with order.precedes

#seq<=1 => seq, merge(sort (seq# (#seq/2)),

sort (seq<<(#seq/2)))

and merge(t1, t2) =

{

newTable(#t1+#t2, choose)

where p1 = 0

and p2 = 0

and choose(i) =

if p1==#t1 then { p2:=p2+1; v } where v=t2 p2 else

if p2==#t2 then { p1:=p1+1; v } where v=t1 p1 else

let v1=t1 p1 and v2=t2 p2

in

if order.precedes(v1, v2) then

p1:=p1+1;

v1

else

p2:=p2+1;

v2

fi;

fi

}

end

and descendingsort = sorting( module precedes(v1, v2) = v1>v2 end )

and ascendingsort = sorting( module precedes(v1, v2) = v1<v2 end )

1.8 Objects

An object is a record of which one or more fields are procedures.
ObjectsProcedures defined within records are called methods, and within the body of a method, the

special symbol this denotes the record itself.

Here we define a procedure rat which constructs objects that represent rational numbers
that can be multiplied. Such objects have a method mul that multiplies its own object with
another rat, and a method add that adds another rat. They also have an equivalence method
and a _print()method.

>> let rat(n, d) =

{| num = n;

den = d;

mul other = rat(this.num*other.num, this.den*other.den);

add other = rat(this.num*other.den+this.den*other.num, this.den*other.den);

_print() = "%/%"%(this.num, this.den);

this === other = this.num*other.den == this.den*other.num

|};

>> rat(6,8);

6/8

If an object has a print() method, then that method is invoked by the system whenever it
has to generate a printable view of the object – the system generates a printable view of value
it returns instead of using the object’s default printable view.

If an object has an equivalence method, (defined by this===other=...) then its equivalence
with other values is decided by that method, rather than by the default structural equivalence
for records.

A method is called in the same way as any other procedure; of course it has to be selected
from its record first:

>> rat(6, 8).mul(rat(1,2));

6/16

>> rat(6, 8).mul(rat(1,2))===rat(3,8);

true

When a method, defined by M
ethod

Calls
method(arg1, ...) = body

is invoked by
object.method(param1, ...)

then the body of the method is evaluated, with this bound to (the value of) object, and arg1, ...
bound to (the values of) param1, ....

Thus, the expression
rat(6, 8).mul(rat(1, 2))

is evaluated by evaluating

rat(this.num ∗ other.num, this.den ∗ other.den)

in a context where this = rat(6, 8), and other = rat(1, 2). In this context, this.num = 6, this.den =
8, other.num = 1, other.den = 2, so the result is a new object constructed by rat(6, 16).

It is not necessary to use this. to qualify fieldnames within methods defined in objects: any
this

Unnecessary

field f of an object referred to within the body of a method of that object is transformed, at
compile-time, into this.f . So, for example, the methods mul and add defined earlier could
have been written:

mul other = rat(num*other.num, den*other.den);

add other = rat(num*other.den+den*other.num, den*other.den);

Our original representation of rationals is somewhat inadequate (no subtraction or division),
Inheritance

and we are not entirely happy with the way rationals are printed. We could, of course, simply
rewrite the definition, but there is a more convenient way that allows us to exploit our original
definition.

We will improve the representation in stages. First we use a record extend on the result of rat
in order to yield a record whose print method presents the number in its simplest form.
The new print method overrides the old print method in the result of ratio.

>> let ratio(n, d) =

rat(n, d) ++ {| _print() = "%/%"%(norm(this.num, this.den));

norm(n, d) = (n/g, d/g) where g = gcd(n, d);

|};

>> ratio(8,16);

1/2

All well and good? Unfortunately not! The arithmetic methods still yield rat results – and
they have their own way of printing.

>> ratio(8,16).add(ratio(8, 16));

256/256

28



This can be fixed by overriding the rat method that is used to construct the results of the
arithmetic operations:

>> let ratio(n, d) =

rat(n, d) ++ {| _print() = "%/%"%(norm(this.num, this.den));

norm(n, d) = (n/g, d/g) where g = gcd(n, d);

rat(n, d) = ratio(n, d);

|};

>> ratio(1,2).add(ratio(8,16));

1/1

Of course it would be better to keep our numbers in their simplest form, and the following
definition accomplishes this:

>> let ratio(n, d) =

rat(n, d) ++

{| _print() = "%/%"%(this.num, this.den);

norm(n, d) = (n/g, d/g) where g = gcd(n, d);

rat(n, d) = ratio(norm(n, d));

|};

Notice that the tuple (this.num, this.den)mentions the names of fields that are not defined
within the record where they are used. Although the interpreter will warn us of this, it is not
a problem here. Within the body of print, the variable this denotes the record from which
it was selected: a record equivalent to

{| num = ...;

den = ...;

mul other = rat(this.num*other.num, this.den*other.den);

add other = rat(this.num*other.den+this.den*other.num,

this.den*other.den);

this === other = this.num*other.den == this.den*other.num

_print() = "%/%"%(this.num, this.den);

norm(n, d) = (n/g, d/g) where g = gcd(n, d);

rat(n, d) = ratio(norm(n, d));

|}

It can be very convenient to define infix and prefix operators on types that are represented
Operators

by records. Here, for example, is a final definement of rat that uses infix notation for rational
addition and multiplication, and prefix - and / for negation and reciprocation, and then
uses negation and reciproaction in the implementation of subtraction and division. The first
parameter of infixes (and the only parameter of prefixes) defined within records must be
this.

>> let rational(n, d) =

ratio(n, d) ++

{|

- this = rational(-num, den);

/ this = rational(den, num);

this * other = mul(other);

this + other = add(other);

this - other = this + (-other);

this / other = this * (/other);

rat(n, d) = rational(n, d);

|};

>> let half=rational(1,2);

>> for n in [half+half, half*half, -half,

half+(half*half), (half+(half*half))/(half*half)] do println n;

4/4

1/4

-1/2

3/4

3/1

It is often necessary to refer to the left-hand operand of an object or record extend from
superwithin the right hand operand. This occurs so frequently when constructing objects, that the

notation
rec1 with rec2

may be used as a shorthand for

let super=rec1 in super++rec2

For example

>> let rec = {| num=5; den=6 |} with {| previous=super; num=6; den=7 |};

>> rec;

{| den=7; num=6; previous={| den=6; num=5 |} |}

2 Obol Types

The details of Obol’s predefined types are given in tabular form in appendix B. In this section
we give a discursive account of some of the more important types.

2.1 Simple Values

2.1.1 ?

The completely undefined Obol value is written ?. It cannot even be compared with anything,
and is useful only for giving “don’t care” initial values to variables or the fields of records.

2.1.2 {}

The simplest useful Obol value is {}. Obol phrases that are evaluated solely for their side-
effects yield this value.

2.1.3 Booleans

The Boolean values are true and false. They are equipped with the usual boolean operators.
Conjunction and disjunction have the usual “short-cut” meanings.

2.1.4 Numbers

Real numbers are represented by 64-bit floating-point numbers, in the range (math.minreal,
math.maxreal).

They are written, using the usual Scientific notation, in the form

integralpart.fractionalpart

followed by an optional Eexponent. (For example 3.141593E6 is an approximation to 106π.)

Integers are represented as 64-bit signed fixed-point numbers. They are written

• in octal form, as a sequence of decimal digits, starting with 0.

• in hex form, as a sequence of hex digits, prefixed with with 0x.

• in decimal form, as a sequence of decimal digits that does not start with 0.

All numbers are equipped with the usual arithmetic and relational operators, as detailed in
Appendix B. Results are integers only if both operands are integers, otherwise they are real.

The built-in mathmodule provides a few of the more important real constants and functions

2.1.5 Characters

Characters are encoded in (16-bit) Unicode. They are written as a single ordinary charac-
ter, or a special character designator, between primes. Special character designators are:
\’ \" \r \n \t \f \b \\ or \nnn where nnn is a three-digit octal number.

The characters are equipped with the same relational operators as the numbers, and (in
addition)

forward c1 >> n2 = the character whose unicode code is n2 more than c1
backward c1 << n2 = the character whose unicode code is n2 less than c1
difference c1 − c2 = the difference in unicode codes between c1 and c2

Where ci are characters, and ni are integers

For example, the following functions map characters to their unicode encodings, and
vice-versa.

let unichar n = ’\000’>>n;

let unicode c = c-’\000’;

2.2 Structured Values

2.2.1 Strings

Strings are encoded as (16-bit) Unicode sequences. They are written as a sequence of char-
acters (or special character designators) between double-quotes. The length of a string, s,
is denoted #s. The string itself is treated as a function from the indices 0, 1, ... #s-1 to its
characters. For example:

>> "Hello, World";

"Hello, World"

>> "Hello, World"(0);

’H’

>> "Hello, World"(4);

’o’

Strings are catenated with the ++ operator:

>> "Ob"++"ol";

"Obol"

The following three substring operators are called take, dropleft, and dropright:

>> "Obol"#3;

"Obo"

>> "Obol"<<2;

"ol"

>> "Obol">>3;

"O"

Strings may be composed using the formatting operator, %, as follows:
Form

atting>> "% is % for %"%["foo", "on sale", 30.0]

"foo is on sale for 30.0"

Successive occurences of % in the formatting string are replaced by the string view of the
successive elements of the sequence.

2.2.2 Sequences

A sequence is written as a bracketed, comma-separated, sequence of expressions. The length
of a sequence, s, is denoted #s. The sequence itself is treated as a function from the indices
0, 1, ... #a-1.

For example:

>> let seq = [1, 2, 3, 4, 5];

>> seq(0);

1

>> seq(1);

2

The sequences constructed this way are immutable. Their elements may not be assigned to:

>> seq(2):=5;

Error: 3 is not assignable

Sequences are catenated with the ++ operator:

>> [0]++seq++[6];

[0, 1, 2, 3, 4, 5, 6]

The three subsequence operators are called take, dropleft, and dropright:

>> seq#3;

[1, 2, 3]

>> seq<<2;

[3, 4, 5]

>> seq>>3;

[1, 2]

The built-in function newTable tabulates a function (or function-like value) as a (constant)
sequence:21

>> let facts = newTable(10, f)

>> where f(n) = n==0=>1, n*f(n-1);

>> facts;

[1, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880]

The built-in function newArray tabulates a function (or function-like value) as a sequence of
locations that may be assigned to.

>> let vfacts = newArray(#facts, facts);

>> vfacts(0):=129;

>> vfacts;

[129, 1, 2, 6, 24, 120, 720, 5040, 40320, 362880]

>> let exploded = newArray(#"foobaz", "foobaz");

>> exploded(2), exploded(5) := ’x’, ’t’;

>> string(exploded);

"foxbat";

The newArray function can accept a simple constant (rather than a function) as its second
argument, in which case the constant is replicated the appropriate number of times.

The following built-in functions also construct sequences:

newArrayOf seq === newArray(#seq, seq)

newTableOf seq === newTable(#seq, seq)

2.2.3 Mappings

See 1.4.2.

2.2.4 Records

See 1.5.
21The conditional expression e0 => e1, e2 is equivalent to if e0 then e1 else e3 fi

3 Java Objects

3.1 Introduction

In this section, which should be skipped by people who are not familiar with both Java and OOP
terminology, we outline the interface between Obol and Java.

It is easy to construct an Obol object (a so-called proxy) that represents a Java package,
interface, or class, and thence to construct a proxy for a Java object of that class, apply a static
method of the class, or get hold of a constant defined statically in the class.

In the following example, we acquire a proxy for the package java.awt, thence the class,
java.awt.Frame, then use the class to build (with .new(...)) a proxy for a frame.22

>> let awt = package("java.awt");

>> let Frame = awt.Frame;

>> let myFrame = Frame.new("EXAMPLE FRAME");

>> awt;

java.awt;

>> Frame;

java.awt.Frame

>> myFrame;

java.awt.Frame

[frame0,0,0,0x0,invalid,hidden,layout=java.awt.BorderLayout,

title=EXAMPLE FRAME,resizable,normal]

There is a shortcut if all you want is the frame: the Obol object java is a proxy for the “package
of packages” whose names begin with java, and (by a punning stunt) we are able to interpret
Obol dot notation appropriately.

>> let myFrame = java.awt.Frame.new("EXAMPLE FRAME");

java.awt.Frame

[frame0,0,0,0x0,invalid,hidden,layout=java.awt.BorderLayout,

title=EXAMPLE FRAME,resizable,normal]

When calling methods on Java proxies, Obol translates parameters from Obol to Java objects
in just the right way. It also translates back from Java values to Obol values when there is an
obvious correspondence. Obol proxy objects are made for Java objects, and Obol sequences
(of Obol values) for Java arrays.

>> java.lang.Double.MAX_VALUE;

1.7976931348623157E308

22An Obol proxy for a Java object is printed using the string yielded by the toString() method of the Java
object.

35



>> java.awt.Color.blue;

java.awt.Color[r=0,g=0,b=255]

>> java.math.BigInteger.new("1000000000000000000000001");

1000000000000000000000001

>> TYPE(java.math.BigInteger.new("1000000000000000000000001"));

‘javaobject "java.math.BigInteger"

>> java.math.BigInteger.new("1000000000000000000000001").bitLength();

80

Perhaps the most useful part of all this,23 is that we can extend Java objects with Obol
methods, and we can build Obol implementations of Java interfaces.

In the following example we define a function that takes an Obol procedure f, and returns an
Obol proxy for a (Java) implementation of the Java interface java.awt.event.ActionListener.
The Java implementation invokes the Obol method actionPerformedwhen its own actionPerformed
method is called, and this invokes f (with the name of the button’s command).

let actionFunction(f) =

java.awt.event.ActionListener.new

{|

actionPerformed(event) = f (event.getActionCommand())

|};

The newmethod of (the Obol proxy for) a Java interface expects a record whose methods have
names identical to those specified by the Java interface, and yields an Obol proxy for a (Java)
implementation of the Java interface.

In the Java awt a button is associated with an action by adding to the button an ActionListener
whose actionPerformedmethod implements the action.

Below we take a slightly different approach by defining an Obol procedure Button. It builds
an Obol button object that can be used as an awt component. The button object has a
withFunction method that arranges for its argument to be called with an appropriate string
argument whenever the button is pressed.

let Button(text) =

java.awt.Button.new(text) with

{| withFunction(f) =

{ super.addActionListener(actionFunction(f));

this

}

|};

23and certainly the most challenging part to implement!

The result of Button is an object composed of (a proxy for) a java.awt.Button extended with
an Obol class that implements withFunction, by calling the addActionListenermethod in the
kernel (super) object, then returns the complete extended object.

We can use these tools to build a (very simple) user interface that simply reports the “press”
events on each of four buttons.

let pressed command = println(command) in

let frame = java.awt.Frame.new("Application") in

{ frame.setLayout(java.awt.GridLayout.new(2,2));

for name in ["Mumble", "Grumble", "Rhubarb", "Crumble"] do

frame.add(Button(name).withFunction(pressed));

frame.pack();

frame.show()

}

Later in this section we give a partial example of how one might build a convenient interface
to the Java Abstract Windowing Toolkit.

Some Java classes and/or objects have fields or methods whose names happen to be Obol
Reserved

M
ethod Nam

es

reserved words. For example, the Matcher class has method end(). To access such a method
simply quote its name as an Obol string. For example

let m = pattern.matcher(string) in m."end"();

Notice that a Button object is composed of a Java core object extended with an Obol ob-
Flyin the

Ointm
ent!

ject. Such Obol extensions of Java objects are called Hybrid objects, and have the important
limitation that calls from the Obol object to the core Java object must be super-calls.

Hybrid objects present two faces: one to the Java world and one to the Obol world.

When either their Java or their Obol methods are invoked from Obol, the right thing happens
– overriding of Java methods by Obol methods works as it would in a non-hybrid object.

But the methods of a hybrid that are normally invoked from Java24 are always the methods
of the Java core object. In short, one can extend a Java object in Obol for use by Obol, but one
cannot do so for use by Java.

This might seem to strike the death-knell for graphical objects, since they are nearly always
implemented by overriding the paint method of something like a Canvas. Well, nearly!
Happily it is a very simple matter to implement a suitable Java adapter class.

24i.e. by Java code written without special knowledge of the Obol machinery.

Here, for example, is the essence of the definition of obol.Canvas (from the inbuilt Java
library).

package obol.awt;

import java.awt.*;

public class Canvas extends java.awt.Canvas

{

Painter p;

public Canvas setPainter(Painter p) { this.p=p; return this; }

public void paint(Graphics g)

{ if (p!=null)

p.paint(g);

else

raise new Error("obol.canvas without Painter");

}

}

An obol.Canvas paints with an obol.Painter

package obol.awt;

import java.awt.*;

public interface Painter { public void paint(Graphics g); }

Next we construct a very small Obol application that illustrates one way of constructing and
using a canvas. First we define a Canvas procedure whose argument should be a procedure
of type java.awt.Graphics->{} that does the painting.

let owt = package("obol.awt")

in let Painter paintOn =

owt.Painter.new {| paint gr = paintOn gr |}

and Canvas paintOn =

owt.Canvas.new().setPainter(Painter paintOn);

The application builds a fixed-size canvas, on which it writes ... the usual shibboleth!

let app = java.awt.Frame.new("Hello World")

and hello = Canvas paintOn where paintOn g =

{ g.setColor(java.awt.Color.RED);

g.setFont(java.awt.Font.decode("sanserif-bold-20"));

g.drawString("Hello, World", 20, 20)

}

in { app.add(hello);

hello.setSize(200, 30);

app.pack();

app.show()

}

3.2 Extracts from a Windowing Toolkit

Below we show an extract from the Obol windowing toolkit.

let owt =

module

let obol = package "obol"

and Painter = obol.awt.Painter.new

and Canvas arg =

obol.awt.Canvas.new arg with

{|

withFunctions obj =

/*

withFunctions

{| mousePressed button (x, y) = ...;

mouseReleased button (x, y) = ...;

mouseMoved button (x, y) = ...;

mouseDragged button (x, y) = ...;

|}

Associates the given methods with the appropriate

events on the underlying canvas. Other events

on the canvas can still be listened for.

*/

let funs = mouseFunctions obj in

{ super.addMouseListener(mouseListener(funs));

super.addMouseMotionListener(mouseMotionListener(funs));

this

};

withPainter obj = { super.setPainter(obj); this };

|}

and mouseFunctions obj =

/*

Yields an obol object, intended to be the argument for

mouseListener or mouseMotionListener, (or both) that specifies

responses to Pressed, Released, Moved, and Dragged events,

and ignores all others.

*/

let mouseResponse f e =

let p = e.getPoint() and b = e.getButton()

in f b (p.x, p.y)

in

{| mousePressed e = mouseResponse (obj.mousePressed) e;

mouseReleased e = mouseResponse (obj.mouseReleased) e;

mouseMoved e = mouseResponse (obj.mouseMoved) e;

mouseDragged e = mouseResponse (obj.mouseDragged) e;

invoke_ e = {}

|}

and reportObj = {| invoke_ it = stderr.println it |}

and actionListener = java.awt.event.ActionListener.new

and actionFunction f = actionListener

{| actionPerformed e = f(e.getActionCommand()) |}

and mouseListener = java.awt.event.MouseListener.new

and mouseMotionListener = java.awt.event.MouseMotionListener.new

and Frame(title) = java.awt.Frame.new(title)

and Label(text) = java.awt.Label.new(text)

and Button(text) = java.awt.Button.new(text) with

{| withFunction(f) = { super.addActionListener(actionFunction(f)); this } |}

end;

Here is a small application that uses the library on the previous page. It makes a rectangle
jump to the cursor when the mouse button is pressed, and follow the cursor when the mouse
is dragged.

import owt;

let MouseCanvas(w, h) = import math in

let x = 0

and y = 0

and canvas = Canvas()

and functions =

{|

mouseDragged b (_x, _y) = { x, y:=_x, _y; canvas.repaint(); };

mousePressed b (_x, _y) = { x, y:=_x, _y; canvas.repaint(); };

mouseReleased b (_x, _y) = { };

mouseMoved b (_x, _y) = { };

|}

and painter = Painter

{| paint g =

{

g.setColor(java.awt.Color.RED);

g.fillRect(x, y, 50, 50);

g.setColor(java.awt.Color.BLACK);

g.drawLine(0, 0, x, y);

g.drawLine(x, y, x+50, y+50);

g.drawLine(canvas.getWidth(), canvas.getHeight(), x+50, y+50);

}

|}

in

{ canvas.withPainter painter;

canvas.withFunctions functions;

canvas.setSize(w, h);

canvas.createBufferStrategy(1);

canvas

};

let app = Frame("Mouse Tracker")

in { app.add(MouseCanvas(600, 400));

app.pack();

app.show();

app.setLocation(200, 200)

}

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

A Note on Performance
Computational performance of the Obol interpreter as a whole is between 60% and 80% of

Perform
ance

the speed of compiled Python, without our having tried to optimize the interpreter in any
way. It is also set to improve dramatically when we add “just-in-time” compilation. But it
has to be said that Obol-to-Java (and Java-to-Obol Proxy) calls are not particularly fast25 This
is because of the intrinsic limitations of the Java reflection machinery that has to be used to
implement such cross-language calls. Despite this speed limitation, cross-language calls are
perfectly satisfactory for responding to java.awt events at user-interface speeds.

We expect that the main use to which the Obol to raw-Java interface will be put will be as a
“glue” language – joining visual interfaces (also composed in Obol) to the applications they
control. The applications themselves can be built in pure Java, or a mixture of Java and Obol:
code with high performance requirements can migrate into Java as it is identified.

25Unless the Java side code uses the Obol extension machinery (which has yet to be documented).

A Built-in Modules

The following tables describe the constants, functions, and nested modules defined by Obol’s
built-in modules. The globalmodule is special because its named values are always in scope
(except in regions of program text where a nested declaration “masks” them). Moreover any
declarations made at the top level of the interactive system are incorporated in it.

Module Name and Type Description
global filter:

(val->bool, seq)->seq

filter(p, xs) is the subsequence of xs whose ele-
ments satisfy p. (mutable if xs is mutable)

global isRecord:

(val, (name:string)*)->bool

is the val a record with fields that include
name1, ...namen

global loadFile: (name:string)->{} loads the file at the given path. IfExpression the
path is not absolute, and this function has been
invoked while loading from another file, then the
path is prefixed with the parent path of the other
file before loading.
IfExpression this prefixed path does not denote
a file, then the path is prefixed with each of the
elements of the Obol search path (see OBOLPATH)
in turn until a file is found or there are no more
search path elements.

global loadModule:

(path:string)->module

loads (if it is not already loaded) the module
whose source text is in the file at the given path.
Modules are cached, so code will not be loaded
more than once unless it has been unloaded (see
unloadModule).
IfExpression the path is not absolute, and this
function has been invoked while loading from an-
other file, then the path is prefixed with the parent
path of the other file before loading.
IfExpression this prefixed path does not denote
a file, then the path is prefixed with each of the
elements of the Obol search path (see OBOLPATH)
in turn until a file is found or there are no more
search path elements.

global map:

(val->val, seq)->seq

map(f , xs) === [f (xs(1), f (xs(2), ...] (mutable if xs is
mutable)

global newArray:

(size:int, spec:val)->array

Makes an array of the given size. IfExpression the
spec is not a function then it is replicated size times
to form the array; otherwise the spec is tabulated
for values 0, 1, · · · size − 1

global newFile:

(path:string)->file

Returns a file object corresponding to path. The
file need not necessarily exist.

global newTable:

(size:int, spec:int->val)->seq

Makes a table (immutable sequence) of the given
size that tabulates the spec for values 0, 1, · · · size−
1.

global newTableOf: (spec:seq)->seq Makes a table that tabulates spec for values
0, 1, · · · #spec − 1

global newTuple:

(size:int, spec:int->val)->seq

Makes a tuple (immutable parameter-packet) of
the given size that tabulates the spec for values
0, 1, · · · size − 1.

global.math atan2:

(y:real, x:real)->real

Computes the phase (−π ≤ θ ≤ π) of the polar
representation (r, θ) of the cartesian coordinates
(x, y)

global.system OBOLPATH: string Obol source file search path (’:’- or ’;’-
separated). Initially set as an environment vari-
able or in the shell script that runs the Obol in-
terpreter by -DOBOLPATH=${OBOLPATH-.}. May be
changed while the interpreter is running.

global.system unloadModule:

(path:string)->module

Unloads the module (if any) that was originally
loaded from the given path. Returns the value of
the unloaded module (or null if no module from
the given path was present)

B Obol Types

For any operator ⊕, the expression e1 ⊕ e2 is evaluated by evaluating e1 and e2 then invoking
the ⊕method of the value of e1. In the following table the value of e1 is denoted this and that
of e2 is denoted arg.

Similarly ⊕e is evaluated by evaluating e then invoking the (unary) ⊕method of its value.

B.1 Built-In Types

this Type Method Type. Description
int #: int->range The range representing the sequence

[this, this + 1, · · · this + arg], unless arg is nega-
tive, in which case the range representing the
sequence [this, this − 1, · · · this + arg + 1].

map (): val->val Yields the value of this at arg, if any; otherwise yields
{}.

map ++: map->map Yields a new mapping, m, that is defined at all the
keys that this and arg are defined at, with m(k) =
arg(k) if arg is defined at k, otherwise m(k) = this(k).
The new mapping is mutable if either this or arg are
mutable.

map this(k) := v IfExpression this is mutable, then this will hence-
forth map k to v, and all other keys to their present
value.

record "⊕": val->val (Where ⊕ is any Obol operator symbol or ===)
Provides support for the use of operators at user-
defined types. IfExpression a method of this name,
defined in this record by this ⊕ arg = expr, ex-
ists in this, then this ⊕ arg is implemented by
this."⊕"(arg).

record "⊕ ": ()->val (Where ⊕ is any Obol operator symbol or ===)
Provides support for the use of operators at user-
defined types. IfExpression a method of this name,
defined in this record by ⊕this = expr, exists in
this, then ⊕ this is implemented by this."⊕"().

record ++: record->record Returns a record whose fields are the union of the
fields of this and arg. The value of each field is its
value in arg if it is a field of arg, otherwise it is its
value in this.

record apply: val->val IfExpression apply is defined in a record rec, then
rec(arg) is implemented by invoking rec. apply(arg).

record for:
()->{| hasNext:()->bool;

next:()->bool

|}

IfExpression this method is defined in a record
rec, then for x in rec do ... is implemented by
let it = rec. for() in
while it.hasNext() do
let x=it.next() in ...

(where it is a completely fresh variable)
record fordelegate: ()->val IfExpression for() is not defined in a record, but the

record has a fordelegate() method, then the in-
terpretation of for ... do ... is handed over to
the object that is the result of invoking this method.
Although the effect of a fordelegate can be simulated
by a for, the latter is more efficient if the delegate
method returns a structure for which Obol defines
a “natural” iterator.

record hashCode: ->int Provides the hook for custom-calculation of the
hash codes of records and objects. Equivalent
records/objects should yield equivalent hash codes.
(Hashcodes are used only when an object is used as
a key in a table)

record print: ->val Provides the hook for custom-printing of records.
IfExpression print is defined in a record rec, then
rec is printed by printing the result of evaluating
rec. print(). (The process is recursive, and termi-
nates in the presence of cyclic print()-dependencies,
but will not terminate if it encounters a print()
method that doesn’t terminate.)

record type: ->val Provides the hook for custom-definition of the type
of a record. IfExpression type is defined in a record
rec, then the type of rec is determined (once and for
all) by evaluating rec. type(). By convention, the
value of every Obol type is either a constant or a
construction.

record update: val->val->val IfExpression this method is defined in a record rec,
then rec(argl) := argr is implemented by invoking
rec. update(argl)(argr).

seq All the global functions that construct tables (immutable sequences) and ar-
rays (mutable sequences) treat strings as sequences of characters, as do for
loops.

seq Tuples constructed by (e1, ..., en) or () are used only to package parameters that
will be passed to procedures. They behave in other respects as (read-only)
sequences.

seq ++: seq->seq Constructs a new sequence that is the catenation of
this and arg. IfExpression either or both are muta-
ble then the result is mutable, otherwise the result
is immutable. Performance: if both operands are
immutable then catenation is a unit-time operation,
otherwise it takes time proportional to the length of
its result.

seq this(k) := v IfExpression this is mutable, then this will hence-
forth map k to v, and all other indices to their present
value.

seq (): int->val Yields the argth element of the sequence. Perfor-
mance: constant time for a table or array con-
structed using the global sequence constructors or
a sequence display. ForExpression a sequence con-
structed by concatenation, access time is (at worst)
linear in the length of the path from the root of the
“cat-tree” to the non-cat-sequence that contains the
element.

seq ===: seq->bool Yields true iff this and arg have the same number of
elements, and their correspondingly-numbered el-
ements are equivalent. (The equivalence algorithm
will terminate, even if given a cyclic structure, pro-
viding all the user-defined equivalence tests that it
invokes terminate)

string %: seq->string Constructs a new string from this, replacing suc-
cessive occurences of % by the formattable view
of successive elements of the sequence. (The for-
mattable view of most values is the sequence of
characters that would be printed if the value were
printed at the top level, except that the formattable
views of strings and chars are never embedded in
quote-marks.)
IfExpression the % is followed by a number, w, fol-
lowed by ’l’ or ’r’, then the formattable view is left
(right) justified in a background of width w that
consists of ’0’s if the first character of w is ’0’, other-
wise consists of spaces.
IfExpression the % is followed by another %, then
it is replaced in the new string by a single %.

B.2 Extension Types used in the Java interface

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

javaclass Obol proxy for a Java class with the same name. Proxies of this type
are constructed using the dot notation onpackageobjects. Examples:
java.lang.Math

java.awt.Window

package("javax").xml.parsers.SAXParser

javaclass new:

(val*)->javaobject

Construct a new Java object of the class
that this is a proxy for, and return the
Obol proxy for it.

javainterface Obol proxy for a Java interface with the same name. Proxies of
this type are constructed using the dot notation on package objects.
Examples:
java.awt.event.ActionListener

package("javax").swing.table.TableModel

javainterface new:

rec->javaobject

Construct a new Java object that imple-
ments (using the methods of the rec) the
Java interface that this is a proxy for, and
return the Obol proxy for it. The record
need not provide all the methods spec-
ified by the interface, providing it has
a method invoke (methodname, args).
The invoke method will be called, in
place of any unprovided method, its
methodname parameter will be the name
of method invoked, and its args param-
eter will be the arguments provided for
that method.

javaobject method:
(val*)->val

Apply the named method of the Java ob-
ject that this is a proxy for to parame-
ters (converted from Obol to Java form as
appropriate) and return the result (con-
verted from Java to Obol form).

javapackage A value of this type is the Obol proxy for a Java package with the
same name (whether such a package exists or not). Examples:
java.lang

java.awt

package("javax").xml.parsers

(Obol Proxy for)
obol.awt.Canvas

setPainter:

obol.awt.Painter->obol.awt.Canvas

set the Painter object for this Canvas, and
return this Canvas.

(Obol Proxy for)
obol.awt.Canvas

setUpdater:

obol.awt.Painter->obol.awt.Canvas

set the Updater object for this Canvas,
and return this Canvas.

(Obol Proxy for)
obol.awt.Lightweight

act: string->{} invoke the action listeners associated
with this

(Obol Proxy for)
obol.awt.Lightweight

addActionListener:

actionlistener->{}

Associate the given action listener with
this lightweight component.

(Obol Proxy for)
obol.awt.Lightweight

setPainter:

obol.awt.Painter->{}

set the Painter object for this Lightweight
component

(Obol Proxy for)
obol.awt.Painter

paint:

java.awt.Graphics->{}

An Obol implementation of a method
that will draw the content of the com-
ponent for which this will be the painter.

(Obol Proxy for)
obol.awt.Paper

setPainter:

obol.awt.Painter->obol.awt.Paper

Set the Painter object for this Paper ob-
ject, and return this Paper object. Paper
is double-buffered canvas with antialias-
ing switched on by default. (Smoother
than Canvas, geddit?)

C Obol Concrete Syntax

Here we specify, concisely, the language that is acceptable to the Obol parser. Not all sentences
of this language are semantically meaningful. Some restrictions are explained in the body of
the manual; others are explained in the notes that follow.

phrase ::= command

command ::= while expr do command

| for pattern in expr do command

| for pattern in expr while expr do command

| expr

∗︷   ︸︸   ︷
, expr := expr

∗︷   ︸︸   ︷
, expr (See note 1)

| expr (See note 6)

expr ::= conditional
︷          ︸︸          ︷
:typeexpr

︷                          ︸︸                          ︷
where declarations (See note 2)

conditional ::= disjunction

∗︷                                           ︸︸                                           ︷
=> conditional , conditional

disjunction ::= conjunction

∗︷                   ︸︸                   ︷
|| conjunction

conjunction ::= relation

∗︷              ︸︸              ︷
&& relation

relation ::= sum

∗︷          ︸︸          ︷
RELOP sum (See note 3)

sum ::= prod

∗︷                                ︸︸                                ︷
ADDOP prod | with prod

prod ::= factor

∗︷                ︸︸                ︷
MULOP factor

factor ::= application

∗︷                        ︸︸                        ︷
EXPOP application

application ::= simple

∗︷                     ︸︸                     ︷
. NAME | simple (See note 6)

| composite

declarations ::= declaration

∗︷                     ︸︸                     ︷
and declaration

commands ::= command

∗︷          ︸︸          ︷
; command

︷︸︸︷
; (See note 6)

declaration ::= pattern︸       ︷︷       ︸
,

= command (See note 4)

matchcase ::= pattern -> command

pattern ::= expr (See note 5)

simplepat ::= simple (See note 5)

typeexpr ::= expr (See note 2)

simple ::= CONST

| NAME

| ( expr︸  ︷︷  ︸
,

)

| { commands }

| [ expr︸  ︷︷  ︸
,

]

| {| declaration︸              ︷︷              ︸
;

︷︸︸︷
; |} (See note 6)

| [| expr -> expr︸                 ︷︷                 ︸
,

|]

| module

︷︸︸︷
let declarations end

| if expr then commands
∗︷                                       ︸︸                                       ︷

else if expr then commands︷                 ︸︸                 ︷
else commands

fi

composite ::= ANYOP application

| try expr then command

| try expr catch matchcase

∗︷                ︸︸                ︷
or matchcase

︷                ︸︸                ︷
then command

| let declarations

︷            ︸︸            ︷
in command

| import expr

∗︷         ︸︸         ︷
and expr

︷            ︸︸            ︷
in command

| case expr of matchcase

∗︷                ︸︸                ︷
or matchcase

| \ pattern︸       ︷︷       ︸
,

-> command

Notes

1. The notation

∗︷︸︸︷
· · · means zero or more occurences of · · ·. The notation

︷︸︸︷
· · · means zero or

one occurences of · · ·. The notation
· · ·︸︷︷︸

punct
means a sequence of one or more occurences of · · ·,

separated by the punct symbol. The parser takes the “longest possible sequence” interpretation
of all optional and repeatable forms.

2. The symbol -> is treated as an ADDOP when parsing type expressions. Type annotations are
presently ignored by Obol’s semantic processors.

3. The relation expression e1 R1 e2 R2 e3 R3 · · · is short for (e1 R1 e2)&&(e2 R2 e3)&&(e3 · · ·)

4. The function/method declaration NAME simplepat simplepat ... = command is syntactic
sugar for NAME = \simplepat -> \simplepat -> ... command

5. An expression is only semantically acceptable as a pattern if it has one of the forms listed in the
left hand column of the table in Appendix E.

6. Obol has a forgiving attitude to semicolons. This means that that the semicolons can (usually)26

be omitted between the commands of a command sequence and between the declarations of a
declaration sequence in a record object.
The price to be paid for this is that all expressions are taken to end at the end of a line unless
the last symbol on that line is a prefix or infix operator, or the first symbol on the following line
is an opening bracket of some kind, i.e. one of the symbols ( { {| [ [|.
For example, the following program is written in Obol’s forgiving syntax.

let gcd(m, n) =

{ while m!=n do if m<n then n:=n-m else if n<m then m:=m-n fi

m

}

let rat(n, d) =

let f = gcd(n, d) in

{| num = n/f

den = d/f

mul(other) = rat

(this.num*other.num, this.den*other.den)

_print() = "%/%"%(num, den)

|}

let a = rat(1,2)

let b = rat(4,5)

let _ = println(a.mul b)

26The only time when a semicolon is required between a command/declaration and the following com-
mand/declaration is when the second of them starts with an opening bracket of some kind.

CONST ::= NUMBER | STRING | CHAR | ‘NAME

RELOP ::= < | <= | == | === | != | !==

| > | >= | ∼= | ∼== | ! | ∼

ADDOP ::= + | - | & | vertical bar | -> | ∼>

| >> | << | ˆ | ++

MULOP ::= * | / | % | <> | ><

ANYOP ::= ADDOP | MULOP | RELOP | EXPOP

EXPOP ::= # | ## | **

NAME ::= A sequence of letters, digits, or underscores, starting with
a letter or underscore

(See note 1)

| ?

NUMBER ::= A real number: a sequence of one or more digits, followed
by a period then zero or more further digits, possibly fol-
lowed by the letter E and one or more further digits.

| A decimal integer: a sequence of decimal digits that does
not begin with 0.

| An octal integer: a sequence of decimal digits that begins
with 0.

| A hexadecimal integer: a sequence of hexadecimal digits
prefixed with 0x.

STRING ::= "

∗︷       ︸︸       ︷
CHARDES "

CHAR ::= ’CHARDES ’

CHARDES ::= any single character except \, " or ’
| \’ apostrophe
| \" doublequote
| \\ backslash
| \r return
| \n newline
| \t tab
| \f formfeed
| \b backspace
| \three octal digits

Notes

1. The names ?, this, super, outer, global are treated in most respects like ordinary vari-
ables, but may not be re-bound in patterns, or as formal parameters of functions or methods.
The name is the “don’t care” pattern within programs. At the top-level of the interpreter it
gets bound to the value of the last phrase to be successfully evaluated.

D Gotcha!

In this section we outline some frequently-occuring circumstances that may give rise to
mystifying reports.

D.1 Syntactic

1. Omitting Semicolons Adjacent commands in a sequence of commands (or declarations
in a sequence of declarations) can be separated by newlines instead of by semicolons. A
command can be split across lines if the last symbol on the first line is an operator (infix
or prefix) or the first symbol on the second line is an opening bracket or parenthesis of
some kind. In the latter case Obol assumes that the first expression on the second line
is the parameter of an application of a method/function/procedure, and warns about a
possible missing semicolon.

Example:

>> let f x = x;

>> { f

(3)

}

Warning: possible omitted semicolon: (location in the source text)

2. Where did my where go? The expression expr where declarations is syntactic sugar
for let declarations in expr. When the dynamic context of an error is being printed,
the “unsugared” expression is shown – and you see a let clause rather than what was
originally written.

>> f(x where x=4);

Error: variable undeclared: f

at f(let x=4 in x)

3. Application is more binding than Selection The expression f x.y is parsed as (f (x)).y.
If you intend to apply f to x.y make sure you parenthesise the operand, writing: f (x.y).

4. What happened to my declaration? Do not use body-less declarations or imports
within a sequence expression. They have no effect.

>> {let x=3; ...; x}

Error: x variable undeclared: x

Solution: shift the declaration(s) to outside the sequence.

>> let x=3 in { ...; x}

D.2 Semantic

1. Attempting to redefine a built-in variable

If you attempt to redefine a built-in variable, the system will stop you.

>> let cat = "cat"; println cat;

Error: Cannot re-bind a global variable: cat to "cat"

The solution is to restrict the scope of the redefinition to a module, and either qualify
the name of the variable with the name of the module, or import the module.

>> let m = module let cat = "cat" end;

>> println (m.cat);

cat

>> import m in println cat;

cat

Note that importing such a module “at the top level” with a body-less import also
counts as a redefinition.

>> import m;

Error: Cannot re-bind a global variable: cat to "cat"

This should not be too troublesome, since body-less imports can always be dispensed
with.

2. Accidentally iterating over a downward interval

This function is intended to be an array bubblesort

let sort xs =

{ for i in 0 # (#xs-1) do

for j in i+1 # (#xs-i-1) do

if xs i > xs j then xs i, xs j := xs j, xs i fi;

xs

}

If #xs==0, then 0 # (#xs - 1)means [0] and the program will cause a subscript error.

There are various solutions, the simplest of which is case analysis on xs.

let sort xs = case #xs of 0->[] or _-> ... (as before) ...

E Pattern Matching

In the following rules v, v1, ...,w,w1, ... denote variables; p, p1, ... denote more general patterns;
k, k1, ... denote values in general; c, c1, ... denote simple values (numbers, strings, characters,
constructors). In the bindings column, p : k means “the bindings resulting from the match of
p to k.”

Pattern Value Succeeds if and only if Bindings
v k always v to k

k always None
c c always None
() () always None
[] [] always None
{} {} always None

‘tag p ‘tag k p matches k p : k
(p1, · · · pn) (k1, · · · km) m = n and all pi match ki all of pi : ki
[p1, · · · pn] [k1, · · · km] m = n and all pi match ki all of pi : ki

{|v1 = p1; · · · vn = pn|} {|w1 = k1; · · ·wm = km|}

{v1, · · ·} ⊆ {w1, · · ·}
and every pi matches

the correspondingly-labelled kj

all of pi : kj

The same variable may occur repeatedly in a structured pattern, but the pattern will only
match the corresponding structured value if all occurences of the pattern match equivalent
(not necessarily identical) values in the structure.

Examples

>> let s=1 in s;

1

>> let k = (1,2) in let (s,t)=k in (s, t);

(1, 2)

>> let k = (1,2) in let (s,s)=k in (s, s);

Error: s is already bound to a value that does not match 1

>> let k = (1,1) in let (s,s)=k in (s, s);

(1, 1)

>> let k = {| a=1; b=(2,3); c=4 |} in case k of {| a=s; b=t |} -> (s, t);

(1, (2,3))

>> let k = {| a=1; b=(2,3); c=4 |} in case k of {| a=s; c=t |} -> (s, t);

(1, 4)

>> let k = ‘record {| a=1; b=(2,3); c=4 |} in case k of ‘record {| a=s; c=t |} -> (s, t);

(1, 4)

>> let k = ‘record {| a=1; b=‘tuple(2,3); c=4 |} in case k of ‘record {| a=s; b=‘tuple t |} -> (s, t);

(1, (2,3))

>> let k = ‘record {| a=1; b=‘tuple(2,3); c=4 |} in case k of ‘record {| a=s; b=t |} -> (s, t);

(1, ‘tuple(2,3))

The following example, uses “unordered pair” records to demonstrate the significance of our
use of the word equivalent in the description of matching patterns with repeated occurences
of variables.

>> let pair(aa,bb) = {| a=aa; b=bb; this===other = a==other.a&&b==other.b || a==other.b&&b==other.a |};

>> let k1 = pair(1,2) and k2=pair(2, 1);

>> k1===k2;

true;

>> let s, s=k1, k2 in s;

{| a=2; b=1; ...|}

F The Interactive Obol Window

If the Obol interpreter is started without any arguments, or if it is started with first argument
-w, then it launches a two-panel interactive Obol window that looks something like this:

Obol phrases are edited in the editing panel (the topmost panel when the system starts), and
when Enter is pressed at the end of the text in the panel that text is sent to the Obol interpreter.
If the text formed a complete Obol phrase then it is evaluated and the resulting output is
appended to the log panel.

The prompt “>>” is shown just above the top panel when the system is ready to read a new
phrase. It is replaced by “>>>” when the system has read part, but not all, of a phrase, and
by “Evaluating” while the system is executing a phrase.

The Up and Down buttons (and the corresponding keys) can be used to navigate in the history
of the session.

Pressing the Interrupt button (or typing Control-C in the top panel) will interrupt the
evaluation of the phrase that is currently being evaluated. This feature causes problems on
some installations, so it has to be enabled by the Interruptable checkbox on the Filemenu.

The Restart button on the Filemenu starts a more-or-less completely new Obol interpreter.

The Quit button on the Filemenu closes down the window.

G Getting and Installing Obol

G.1 Availability

The Obol installer can be found at

ftp::ftp.comlab.ox.ac.uk/pub/Packages/JAPE/OBOL/ObolInstall.jar

It can be run on a Windows or a Unix machine, providing the java runtime environment has
been installed first.

G.2 Windows installation

1. Ensure that java 1.4 (or later) has been installed on your machine.

2. Acquire (see above) ObolInstall.jar

3. Decide on the folder that will be used as the installation folder for Obol. We recommend
that you make a new folder for this purpose.

4. OPTIONAL: Move ObolInstall.jar to the installation folder.

5. Double click on ObolInstall.jar in the installation folder. This should bring up a new
window that belongs to the installation program.

Unless you moved ObolInstall.jar to the installation directory in the previous step,
you should click on the “Choose Folder” button in that window, and choose the instal-
lation folder.

Click on the Install button in that Window. If all goes well a new icon will appear
in the installation folder. That’s the icon you click on to start the Obol interpreter. You
can move it (or send it) anywhere in your windows filestore.

6. (If all doesn’t go well, then on most Windows systems with java installed it is possible
to start Obol with a doubleclick on the file Obol.jar)

G.3 Unix installation

1. Ensure that java 1.4 (or later) has been installed on your machine, and that the java
binary directory is somewhere on your path.

2. Acquire (see above) ObolInstall.jar

3. Decide on the directory that will be used as the installation directory for Obol.

4. OPTIONAL: Move ObolInstall.jar to the installation directory, and change direc-
tory to the installation directory.

5. Run the command java -jar ObolInstall.jar. This should bring up a new window
that belongs to the installation program.

Unless you moved ObolInstall.jar to the installation directory in the previous step,
you should click on the “Choose Folder” button in that window, and choose the direc-
tory.

Click on the Install button in that Window. If all goes well a new executable file,
obol, will appear in the installation directory.

6. (If all doesn’t go well, then on most Unix systems with java installed it is possible to
start Obol with the command java -jar Obol.jar)

Note 1 (Page 6) Matching Declarations
In general a declaration takes the form pattern = expression – the declaration succeeds,
binding the variables in the pattern, providing that the expression evaluates to a structure
that the pattern matches; it fails, raising a bind exception, if the pattern fails to match.
Pattern matching rules are given in Appendix E

Note 2 (Page 8) Obol and Shell Scripts
You can use the Obol system in “shell scripts”, that can be run as complete programs.

Here’s a Unix/Linux “Hello, World” shell script:

#!/bin/env obol

println("Hello, World");

When executing an Obol program configured as a shell script (i.e. with #! on the first line),
the Obol interpreter automatically supresses the printing of null, so this program will print
the specified text, and nothing more.

The Unix command obol is simply a shell script containing the text

exec java -jar -server path_to_the_Obol_jar_file/Obol.jar $*

The system takes the following arguments, which are inspected in succession from the left

-w Use the interactive window.27

-s Force evaluation in scripting mode.
+s Forbid evaluation in scripting mode.28

-h number Set the log panel buffer size to number lines.
-f filename Input the named Obol source file.29

filename Input the named Obol source file; stop inspecting arguments.30

-- As above, but take input from the terminal.31

When there are no arguments, the system puts an interactive window up.

27If -w appears, it must be the first argument.
28Even if the Obol source files looks like a script.
29The environment variable OBOLPATH is used as a search path when opening Obol source files.
30This and all subsequent arguments are handed to the Obol program in the variable system.args.
31... or from the editing panel, if the interactive Obol Window is being used.

i



Note 3 (Page 14) Obol Exceptions
Although the simplest exceptions are represented as strings, Obol permits the throwing of
other types of exception. These are delivered to the catch clause as a construction labelled
with the data constructor ‘thrown.

>> try throw 45 catch s -> s;

‘thrown 45

>> try throw ([3,5,7], 45) catch s -> s;

‘thrown ([3,5,7], 45))

If uncaught they are reported in the following form:

>> throw ([3,5,7], 45);

Error: thrown

at throw([ 3, 5, 7 ], 45)

Subscript errors are normally reported in this form

>> [1,2,3](4);

Error: subscript

at [ 1, 2, 3 ](4)

but they can be caught. They are delivered to the catch clause as a construction labelled
with the data constructor ‘subscript.

>> try [1,2,3](4) catch s -> s;

‘subscript ([1, 2, 3], 4)

It is possible to discriminate on the content or structure of an exception, as the following
examples illustrate:

>> let f x =

case x of

1 -> throw "one"

or 2 -> throw ["two"]

or 3 -> [1](2);

>> try f 1 catch ‘thrown x -> x or other->other;

"one"

>> try f 2 catch ‘thrown x -> x or other->other;

["two"]

>> try f 3 catch ‘thrown x -> x

or ‘subscript (s, i) -> 1968

or other->other;

1968

Running out of cases in a case expression causes a case error to be thrown. Here’s an
example.

>> try f 0 catch x -> x

‘case ("case x of 1 -> throw("one") or

2 -> throw["two"] or

3 -> ([1])(2) no cases match 0", 0)

Note 4 (Page 22) Records and Keyword Parameters
Records and record-patterns can be used to give the effect of “keyword-parameters”:

>> let subst {| pat = regex; all = a; repl = r |} =

let p = java.util.regex.Pattern.compile regex

in if a then

\target -> p.matcher(target).replaceAll r

else

\target -> p.matcher(target).replaceFirst r

fi;

>> subst {| pat="foo"; all=true; repl="baz" |} "foobaz is foo for you";

"bazbaz is baz for you"

>> subst {| pat="foo"; all=false; repl="baz" |} "foobaz is foo for you";

"bazbaz is foo for you"

Perhaps less obviously, record-extension can be used to give the effect of keyword
parameters with defaults:

>> let substitute params = subst ({| all=true; repl="" |} ++ params);

>> substitute {| pat="foo" |} "foobaz is foo for you";

"baz is for you"

Note 5 (Page 22) Record Equivalence
In this note we explain why Obol supports record-specific definitions of record equivalence.

Suppose we set out to represent rational numbers with records. It would make sense to
build a rational-constructor procedure:

>> let rat(n, d) = {| num=n; den=d |};

It may or may not make sense for the implementer to insist that all rationals be kept in
normal form (numerator and denominator coprime), but it is the implementer’s decision.
Were she to decide (perhaps for reasons of efficiency) not to do so, then the structural
equivalence between rational representations would not concide with the semantics of
rational numbers. For, although rat(2, 4) and rat(4, 8) represent identical rationals, they are
not structurally equivalent.

>> rat(2,4)===rat(4,8);

false

This isn’t really satisfactory – we want the equivalence test to be a semantic equivalence test.
Happily Obol permits the designer of a record-based data type to define an equivalence
procedure for that type. Anticipating, to some extent, our introduction to Objects, we give a
glimpse of how this can be done for a (non-normalized) representation of rationals.

>> let rat(n, d) =

{| num = n;

den = d;

this === other = this.num*other.den == this.den*other.num

|};

>> rat(2,4)===rat(4,8);

true

>> rat(2,4)===rat(5,8);

false

Records representing rationals are now associated with an equivalence procedure, ===, that
can be used as an infix operator between them and any other record with num and den fields.

Note 6 (Page 22) Structural Equivalence
Obol uses a structural equivalence procedure that is powerful enough to decide the
equivalence of two finite, but posssibly cyclic, data structures – returning true if and only if
the structures are isomorphic. The isomorphism is determined under the assumption that
identically-named methods in distinct records are isomorphic if they have identical bound
variables and bodies.32

In the following example ones1 and ones2 are distinct but isomorphic cyclic structures –
representing an infinite list of 1s.

>> let ones1 = {| hd=1; tl={} |}; ones1.tl:=ones1;

>> ones1;

{| hd=1; tl=... |}

>> let ones2 = {| hd=1; tl={| hd=1; tl={} |} |}; ones2.tl.tl:=ones2;

>> ones2;

{| hd=1; tl={| hd=1; tl=... |} |}

>> equivalent(ones1, ones2);

true

In the following example, we define the core of a list-handling package, in which list
equivalence is (almost)33 structural equivalence.

32This assumption is necessary because (in general) the isomorphism of two procedures is undecideable.
33We insist on nil being a unique empty record, not just any empty record.

>> let cons(h, t) =

{| hd=h;

tl=t;

|}

and nil =

{|

this === other = other==nil

|};

>> cons(1, cons(2, nil))==cons(1, cons(2, nil));

false

>> cons(1, cons(2, nil))===cons(1, cons(2, nil));

true

>> nil==={||};

false

>> {||}===nil;

false

$ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $ $

It is easy to experiment with tangled list structures. Below we define a function that
replaces all occurences of {} in a cons-built structure with the root of the structure and
construct a few tangled structures with it.

>> let tangle root = { weave true root; root }

where weave top cell =

if cell==root && !top then

{}

else

if isRecord(cell, "hd", "tl") then

if isNull(cell.hd) then cell.hd:=root else weave false (cell.hd) fi;

if isNull(cell.tl) then cell.tl:=root else weave false (cell.tl) fi;

fi;

>> let a = tangle(cons(1, cons(2, cons(3, cons(1, cons(2, cons(3, {})))))))

and b = tangle(cons(1, cons(2, cons(3, {}))))

and c = tangle(cons(cons({}, {}), cons({}, {})))

and d = tangle(cons({}, {}));

>> a===b;

true

>> c===d;

true

